20

Implementation of One-Dimensional Elements

Implementation of Plane (2D) Bar Element



Degrees of freedom and conjugate forces

$$\mathbf{u}^e = \begin{bmatrix} u_{x1} \\ u_{y1} \\ u_{x2} \\ u_{y2} \end{bmatrix} \qquad \mathbf{f}^e = \begin{bmatrix} f_{x1} \\ f_{y1} \\ f_{x2} \\ f_{y2} \end{bmatrix}$$

Global Stiffness Matrix of Plane Bar

$$\mathbf{K}^e = \frac{EA}{\ell}\begin{bmatrix} c^2 & sc & -c^2 & -sc \\ sc & s^2 & -sc & -s^2 \\ -c^2 & -sc & c^2 & sc \\ -sc & -s^2 & sc & s^2 \end{bmatrix}$$

(Chapter 3)

$$= \frac{EA}{\ell^3}\begin{bmatrix} x_{21}x_{21} & x_{21}y_{21} & -x_{21}x_{21} & -x_{21}y_{21} \\ x_{21}y_{21} & y_{21}y_{21} & -x_{21}y_{21} & -y_{21}y_{21} \\ -x_{21}x_{21} & -x_{21}y_{21} & x_{21}x_{21} & x_{21}y_{21} \\ -x_{21}y_{21} & -y_{21}y_{21} & x_{21}y_{21} & y_{21}y_{21} \end{bmatrix}$$

(form actually implemented)

Reason for the last form: symbolic computation. The length ℓ is irrational function of the node coordinates, and segregation helps simplification.

Mathematica Module to Form Stiffness Matrix of a 2-Node Plane Bar Element

```
PlaneBar2Stiffness[ncoor_,Em_,A_,options_]:= Module[
 {x1,x2,y1,y2,x21,y21,EA,numer,L,LL,LLL,Ke},
  {{x1,y1},{x2,y2}}=ncoor; {x21,y21}={x2-x1,y2-y1};
  EA=Em*A; {numer}=options; LL=x21^2+y21^2; L=Sqrt[LL];
  If [numer,{x21,y21,EA,LL,L}=N[{x21,y21,EA,LL,L}]];
  If [!numer, L=PowerExpand[L]]; LLL=Simplify[LL*L];
  Ke=(Em*A/LLL)*{{ x21*x21, x21*y21,-x21*x21,-x21*y21},
                 { y21*x21, y21*y21,-y21*x21,-y21*y21},
                 {-x21*x21,-x21*y21, x21*x21, x21*y21},
                 {-y21*x21,-y21*y21, y21*x21, y21*y21}};
  Return[Ke]];
```

Test of Plane Bar Element with Numerical Inputs

```
ClearAll[A,Em,L];
ncoor={{0,0},{30,40}}; Em=1000; A=5;
Ke= PlaneBar2Stiffness[ncoor,Em,A,{True}];
Print["Numerical Elem Stiff Matrix: "];
Print[Ke//MatrixForm];
Print["Eigenvalues of Ke=",Chop[Eigenvalues[N[Ke]]]];
Print["Symmetry check=",Simplify[Chop[Transpose[Ke]-Ke]]];
```

Numerical Elem Stiff Matrix:

$$\begin{pmatrix} 36. & 48. & -36. & -48. \\ 48. & 64. & -48. & -64. \\ -36. & -48. & 36. & 48. \\ -48. & -64. & 48. & 64. \end{pmatrix}$$

Eigenvalues of Ke = {200., 0, 0, 0}

Symmetry check={{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

Test of Plane Bar Element with Symbolic Inputs

```
ClearAll[A,Em,L];
ncoor={{0,0},{L,0}};
Ke=  PlaneBar2Stiffness[ncoor,Em,A,{False}];
kfac=Em*A/L; Ke=Simplify[Ke/kfac];
Print["Symbolic Elem Stiff Matrix: "];
Print[kfac," ",Ke//MatrixForm];
Print["Eigenvalues of Ke=",kfac,"*",Eigenvalues[Ke]];
```

Symbolic Elem Stiff Matrix:

$$\frac{A\,Em}{L} \begin{pmatrix} 1 & 0 & -1 & 0 \\ 0 & 0 & 0 & 0 \\ -1 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

Eigenvalues of Ke = $\frac{A\,Em}{L} * \{0, 0, 0, 2\}$

Plane (2D) Beam-Column Element in Local System



$$\bar{\mathbf{u}}^e = \begin{bmatrix} \bar{u}_{x1} \\ \bar{u}_{y1} \\ \theta_{z1} \\ \bar{u}_{x2} \\ \bar{u}_{y2} \\ \theta_{z2} \end{bmatrix} \qquad \bar{\mathbf{f}}^e = \begin{bmatrix} \bar{f}_{x1} \\ \bar{f}_{y1} \\ m_{z1} \\ \bar{u}_{x2} \\ \bar{u}_{y2} \\ m_{z2} \end{bmatrix}$$

$$\overline{\mathbf{K}}^e = \frac{EA}{\ell} \begin{bmatrix} 1 & 0 & 0 & -1 & 0 & 0 \\ & 0 & 0 & 0 & 0 & 0 \\ & & 0 & 0 & 0 & 0 \\ & & & 1 & 0 & 0 \\ & & & & 0 & 0 \\ symm & & & & & 0 \end{bmatrix} + \frac{EI}{\ell^3} \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ & 12 & 6\ell & 0 & -12 & 6\ell \\ & & 4\ell^2 & 0 & -6\ell & 2\ell^2 \\ & & & 0 & 0 & 0 \\ & & & & 12 & -6\ell \\ symm & & & & & 4\ell^2 \end{bmatrix}$$

bar contribution   ***Bernoulli-Euler beam contribution***

Plane (2D) Beam-Column Element in Global System



$$\mathbf{u}^e = \begin{bmatrix} u_{x1} \\ u_{y1} \\ \theta_{z1} \\ u_{x2} \\ u_{y2} \\ \theta_{z2} \end{bmatrix} \qquad \mathbf{f}^e = \begin{bmatrix} f_{x1} \\ f_{y1} \\ m_{z1} \\ f_{x2} \\ f_{y2} \\ m_{z2} \end{bmatrix}$$

$$\bar{\mathbf{u}}^e = \begin{bmatrix} \bar{u}_{x1} \\ \bar{u}_{y1} \\ \theta_{z1} \\ u_{x2} \\ \bar{u}_{y2} \\ \theta_{z2} \end{bmatrix} = \begin{bmatrix} c & s & 0 & 0 & 0 & 0 \\ -s & c & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & c & s & 0 \\ 0 & 0 & 0 & -s & c & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} u_{x1} \\ u_{y1} \\ \theta_{z1} \\ u_{x2} \\ u_{y2} \\ \theta_{z2} \end{bmatrix} = \mathbf{T}\,\mathbf{u}^e$$

$$\mathbf{K}^e = \mathbf{T}^T \bar{\mathbf{K}}^e\, \mathbf{T}$$

Mathematica Module to Form Stiffness Matrix of a 2-Node Plane Beam-Column Element

```
PlaneBeamColumn2Stiffness[ncoor_,Em_,{A_,Izz_},options_]:=Module[
 {x1,x2,y1,y2,x21,y21,EA,EI,numer,L,LL,LLL,Te,Kebar,Ke},
  {{x1,y1},{x2,y2}}=ncoor; {x21,y21}={x2-x1,y2-y1};
   EA=Em*A; EI=Em*Izz; {numer}=options;
   LL=Simplify[x21^2+y21^2]; L=Sqrt[LL];
   If [numer,{x21,y21,EA,EI,LL,L}=N[{x21,y21,EA,EI,LL,L}]];
   If [!numer, L=PowerExpand[L]]; LLL=Simplify[LL*L];
   Kebar= (EA/L)*{
  { 1,0,0,-1,0,0},{0,0,0,0,0,0},{0,0,0,0,0,0},
  {-1,0,0, 1,0,0},{0,0,0,0,0,0},{0,0,0,0,0,0}} +
        (2*EI/LLL)*{
  { 0,0,0,0,0,0},{0, 6, 3*L,0,-6, 3*L},{0,3*L,2*LL,0,-3*L,  LL},
  { 0,0,0,0,0,0},{0,-6,-3*L,0, 6,-3*L},{0,3*L,  LL,0,-3*L,2*LL}};
   Te={{x21,y21,0,0,0,0}/L,{-y21,x21,0,0,0,0}/L,{0,0,1,0,0,0},
       {0,0,0,x21,y21,0}/L,{0,0,0,-y21,x21,0}/L,{0,0,0,0,0,1}};
   Ke=Transpose[Te].Kebar.Te;
   Return[Ke]];
```

Test of Plane Beam-Column Element with Numerical Inputs

```
ClearAll[L,Em,A,Izz];
ncoor={{0,0},{3,4}}; Em=100; A=125; Izz=250;
Ke= PlaneBeamColumn2Stiffness[ncoor,Em,{A,Izz},{True}];
Print["Numerical Elem Stiff Matrix: "];
Print[Ke//MatrixForm];
Print["Eigenvalues of Ke=",Chop[Eigenvalues[Ke]]];
```

Numerical Elem Stiff Matrix:

$$\begin{pmatrix} 2436. & 48. & -4800. & -2436. & -48. & -4800. \\ 48. & 2464. & 3600. & -48. & -2464. & 3600. \\ -4800. & 3600. & 20000. & 4800. & -3600. & 10000. \\ -2436. & -48. & 4800. & 2436. & 48. & 4800. \\ -48. & -2464. & -3600. & 48. & 2464. & -3600. \\ -4800. & 3600. & 10000. & 4800. & -3600. & 20000. \end{pmatrix}$$

Eigenvalues of Ke = {34800., 10000., 5000., 0, 0, 0}

Test of Plane Beam-Column Element with Symbolic Inputs

```
ClearAll[L,Em,A,Izz];
ncoor={{0,0},{3*L/5,4*L/5}};
Ke=  PlaneBeamColumn2Stiffness[ncoor,Em,{A,Izz},{False}];
Print["Symbolic Elem Stiff Matrix:"]; kfac=Em;
Ke=Simplify[Ke/kfac]; Print[kfac," ",Ke//MatrixForm];
Print["Eigenvalues of Ke=",kfac,"*",Eigenvalues[Ke]];
```

Symbolic Elem Stiff Matrix:

$$
\mathrm{Em}\begin{pmatrix}
\frac{3\,(64\,\mathrm{Izz}+3\,A\,L^2)}{25\,L^3} & \frac{12\,(-12\,\mathrm{Izz}+A\,L^2)}{25\,L^3} & -\frac{24\,\mathrm{Izz}}{5\,L^2} & -\frac{3\,(64\,\mathrm{Izz}+3\,A\,L^2)}{25\,L^3} & -\frac{12\,(-12\,\mathrm{Izz}+A\,L^2)}{25\,L^3} & -\frac{24\,\mathrm{Izz}}{5\,L^2} \\
\frac{12\,(-12\,\mathrm{Izz}+A\,L^2)}{25\,L^3} & \frac{4\,(27\,\mathrm{Izz}+4\,A\,L^2)}{25\,L^3} & \frac{18\,\mathrm{Izz}}{5\,L^2} & -\frac{12\,(-12\,\mathrm{Izz}+A\,L^2)}{25\,L^3} & -\frac{4\,(27\,\mathrm{Izz}+4\,A\,L^2)}{25\,L^3} & \frac{18\,\mathrm{Izz}}{5\,L^2} \\
-\frac{24\,\mathrm{Izz}}{5\,L^2} & \frac{18\,\mathrm{Izz}}{5\,L^2} & \frac{4\,\mathrm{Izz}}{L} & \frac{24\,\mathrm{Izz}}{5\,L^2} & -\frac{18\,\mathrm{Izz}}{5\,L^2} & \frac{2\,\mathrm{Izz}}{L} \\
-\frac{3\,(64\,\mathrm{Izz}+3\,A\,L^2)}{25\,L^3} & -\frac{12\,(-12\,\mathrm{Izz}+A\,L^2)}{25\,L^3} & \frac{24\,\mathrm{Izz}}{5\,L^2} & \frac{3\,(64\,\mathrm{Izz}+3\,A\,L^2)}{25\,L^3} & \frac{12\,(-12\,\mathrm{Izz}+A\,L^2)}{25\,L^3} & \frac{24\,\mathrm{Izz}}{5\,L^2} \\
-\frac{12\,(-12\,\mathrm{Izz}+A\,L^2)}{25\,L^3} & -\frac{4\,(27\,\mathrm{Izz}+4\,A\,L^2)}{25\,L^3} & -\frac{18\,\mathrm{Izz}}{5\,L^2} & \frac{12\,(-12\,\mathrm{Izz}+A\,L^2)}{25\,L^3} & \frac{4\,(27\,\mathrm{Izz}+4\,A\,L^2)}{25\,L^3} & -\frac{18\,\mathrm{Izz}}{5\,L^2} \\
-\frac{24\,\mathrm{Izz}}{5\,L^2} & \frac{18\,\mathrm{Izz}}{5\,L^2} & \frac{2\,\mathrm{Izz}}{L} & \frac{24\,\mathrm{Izz}}{5\,L^2} & -\frac{18\,\mathrm{Izz}}{5\,L^2} & \frac{4\,\mathrm{Izz}}{L}
\end{pmatrix}
$$

Eigenvalues of Ke = $\mathrm{Em}*\{0, 0, 0, \frac{2A}{L}, \frac{2\,\mathrm{Izz}}{L}, \frac{6\,(4\,\mathrm{Izz} + \mathrm{Izz}\,L^2)}{L^3}\}$

Space (3D) Bar Element



$$\mathbf{u}^e = \begin{bmatrix} u_{x1} \\ u_{y1} \\ u_{z1} \\ u_{x2} \\ u_{y2} \\ u_{z2} \end{bmatrix} \qquad \mathbf{f}^e = \begin{bmatrix} f_{x1} \\ f_{y1} \\ f_{z1} \\ f_{x2} \\ f_{y2} \\ f_{z2} \end{bmatrix}$$

Global Stiffness of 3D Bar Element

$$\mathbf{K}^e = \frac{EA}{\ell^3}\begin{bmatrix} x_{21}x_{21} & x_{21}y_{21} & x_{21}z_{21} & -x_{21}x_{21} & -x_{21}y_{21} & -x_{21}z_{21} \\ x_{21}y_{21} & y_{21}y_{21} & x_{21}z_{21} & -x_{21}y_{21} & -y_{21}y_{21} & -y_{21}z_{21} \\ x_{21}z_{21} & y_{21}z_{21} & z_{21}z_{21} & -x_{21}z_{21} & -y_{21}z_{21} & -z_{21}z_{21} \\ -x_{21}x_{21} & -x_{21}y_{21} & -x_{21}z_{21} & x_{21}x_{21} & x_{21}y_{21} & x_{21}z_{21} \\ -x_{21}y_{21} & -y_{21}y_{21} & -x_{21}z_{21} & x_{21}y_{21} & y_{21}y_{21} & y_{21}z_{21} \\ -x_{21}z_{21} & -y_{21}z_{21} & -z_{21}z_{21} & x_{21}z_{21} & y_{21}z_{21} & z_{21}z_{21} \end{bmatrix}$$

(Derivation in Exercise 6.10)

Mathematica Module to Form Stiffness Matrix of a 2-Node Space Bar Element

```
SpaceBar2Stiffness[ncoor_,Em_,A_,options_]:=Module[
 {x1,x2,y1,y2,z1,z2,x21,y21,z21,EA,numer,L,LL,LLL,Ke},
 {{x1,y1,z1},{x2,y2,z2}}=ncoor;{x21,y21,z21}={x2-x1,y2-y1,z2-z1};
  EA=Em*A; {numer}=options;  LL=x21^2+y21^2+z21^2; L=Sqrt[LL];
  If [numer,{x21,y21,z21,EA,LL,L}=N[{x21,y21,z21,EA,LL,L}]];
  If [!numer, L=PowerExpand[L]]; LLL=Simplify[LL*L];
  Ke=(Em*A/LLL)*
     {{ x21*x21, x21*y21, x21*z21,-x21*x21,-x21*y21,-x21*z21},
      { y21*x21, y21*y21, y21*z21,-y21*x21,-y21*y21,-y21*z21},
      { z21*x21, z21*y21, z21*z21,-z21*x21,-z21*y21,-z21*z21},
      {-x21*x21,-x21*y21,-x21*z21, x21*x21, x21*y21, x21*z21},
      {-y21*x21,-y21*y21,-y21*z21, y21*x21, y21*y21, y21*z21},
      {-z21*x21,-z21*y21,-z21*z21, z21*x21, z21*y21, z21*z21}};
   Return[Ke];
];
```

Test of Space Bar Element with Numerical Inputs

```
ClearAll[A,Em];
ncoor={{0,0,0},{2,3,6}}; Em=343; A=10;
Ke= SpaceBar2Stiffness[ncoor,Em,A,{True}];
Print["Numerical Elem Stiff Matrix: "];
Print[Ke//MatrixForm];
Print["Eigenvalues of Ke=",Chop[Eigenvalues[Ke]]];
```

Numerical Elem Stiff Matrix:

$$\begin{pmatrix} 40. & 60. & 120. & -40. & -60. & -120. \\ 60. & 90. & 180. & -60. & -90. & -180. \\ 120. & 180. & 360. & -120. & -180. & -360. \\ -40. & -60. & -120. & 40. & 60. & 120. \\ -60. & -90. & -180. & 60. & 90. & 180. \\ -120. & -180. & -360. & 120. & 180. & 360. \end{pmatrix}$$

Eigenvalues of Ke = {980., 0, 0, 0, 0, 0}

Test of Space Bar Element with Symbolic Inputs

```
ClearAll[A,Em,L];
ncoor={{0,0,0},{L,2*L,2*L}/3};
Ke=  SpaceBar2Stiffness[ncoor,Em,A,{False}];
kfac=Em*A/(9*L); Ke=Simplify[Ke/kfac];
Print["Symbolic Elem Stiff Matrix: "];
Print[kfac," ",Ke//MatrixForm];
Print["Eigenvalues of Ke=",kfac,"*",Eigenvalues[Ke]];
```

Symbolic Elem Stiff Matrix:

$$\frac{A\,Em}{9\,L}\begin{pmatrix} 1 & 2 & 2 & -1 & -2 & -2 \\ 2 & 4 & 4 & -2 & -4 & -4 \\ 2 & 4 & 4 & -2 & -4 & -4 \\ -1 & -2 & -2 & 1 & 2 & 2 \\ -2 & -4 & -4 & 2 & 4 & 4 \\ -2 & -4 & -4 & 2 & 4 & 4 \end{pmatrix}$$

Eigenvalues of Ke = $\frac{A\,Em}{9\,L} * \{0, 0, 0, 0, 0, 18\}$

Space Beam Element
(Details in Notes & HW Exercises)